# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=network-app-stress-tester
pkgver=19.df75391
pkgrel=2
groups=('blackarch' 'blackarch-dos' 'blackarch-networking' 'blackarch-fuzzer')
pkgdesc='Network Application Stress Testing Yammer.'
arch=('x86_64' 'aarch64')
url='https://github.com/PherricOxide/Network-App-Stress-Tester'
license=('GPL3')
depends=()
makedepends=('git')
source=("$pkgname::git+https://github.com/PherricOxide/Network-App-Stress-Tester.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

build() {
  cd "$pkgname/src"

  g++ fast.c -o $pkgname
}

package() {
  cd $pkgname

  install -Dm 755 "src/$pkgname" "$pkgdir/usr/bin/$pkgname"
  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README
}

